# Maintainer:
# Contributor: Daniel Bermond < yahoo-com: danielbermond >

pkgname=libraqm
pkgver=0.5.0
pkgrel=1
pkgdesc='A library that encapsulates the logic for complex text layout'
arch=(x86_64)
url='https://github.com/HOST-Oman/libraqm/'
license=(MIT)
depends=(freetype2 fribidi)
makedepends=(gtk-doc)
source=(https://github.com/HOST-Oman/libraqm/releases/download/v$pkgver/raqm-$pkgver.tar.gz)
sha256sums=('a06036ad6bd0abdc656de0b4189cc7f397ec1beb17061012732e32761a02c989')

build() {
    cd raqm-$pkgver
    
    ./configure \
        --prefix=/usr \
        --enable-static=no \
        --enable-shared=yes \
        --enable-fast-install=yes \
        --enable-gtk-doc=yes \
        --enable-gtk-doc-html=yes \
        --enable-gtk-doc-pdf=no
    make
}

package() {
    cd raqm-$pkgver
    
    make DESTDIR="$pkgdir" install
    
    install -D -m644 COPYING "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
}
